סיכום תרגול 2
1. INT 21h
INT 21h הינו פנייה לספריית תוכניות שירות של מערכת ההפעלה DOS.  בקורס "מערכות הפעלה" נראה שלסוג כזה של תוכניות יש את השם "קריאות מערכת" ושסיפוק תוכניות שירות כגון אלו הן אחת מעיקרי תפקידיו של מערכת הפעלה. מבחינה מעשית, יישום שרץ תחת מערכת הפעלה (בחייחוד היום), מוציא לפועל את רצונו בעיקר דרך תוכניות שירות כגון אלו.  קוד של יישומים עושים בעיקר קבלת החלטות מה הם רוצים שיתבצע, אבל ברגע שהיישום רוצה לתת ביטוי חיצוני למה שהוא עשה למשל במובן של קלט מהמשתמש או פלט למסך/דיסקים הוא חייב לפנות למערכת. 
פקודת המכונה INT היא פקודת הסתעפות מסוג מיוחד שנקרא "פסיקת תוכנה". כשנבדר על פסיקות נקראה כיצד היא פועלת.  מה שמיוחד בה היא שהמסתעפת מחוץ לקוד של התוכנית הנוכחית.  אגב, Linux משתמש לאותה מטרה ש-DOS מממש ב-INT 21h בפקודה INT 80h.

כאשר אנחנו מסתעפים ל-INT 21h אנחנו מגיעים לקוד סדרן שמפנה את המעבד לרוטינת שירות לפי הערך שהוא מוצא באוגר AH.  יש שם מספר לא מבוטל של רוטינות, בין השאר לפתיחה/קריאה/כתיבה של קבצים, שינוי הספרייה הנוכחית וכו'.   
בקורס מספיק שתכירו 3 מהם:
· אופציה AH=9 שזה הדפס תוכן הזיכרון מהנקודה המוצבעת ע"י DS:DX עד ל-‘$’ הראשון
· אופציה AH=1 שזה קרא מקש ascii בודד מהמקלדת (עם echo) והחזר אותו ב-AL
·   אופציה AH=2 שזה הדפס את תו שקוד ה-ascii שלו ב-DL








2. המלצה לשיטת עבודה בתכנות באסמבלי
תכנות אלגורתם מורכב באסמבלי אינו פיקניק ושיטת עבודה שתחסוך לכם הסתבכויות שניתנות למניעה היא כדלקמן:
· תפתחו את התוכנית בכמה שלבים
· בשלב ראשון תממשו אותו בשפת עילית, רצוי C.
· תתרגמו ידנית את התוכנית ב-C לאסמבלי במודל small:
1. תבנו את המסגרת של התוכנית.
2. תממשו ב-.DATA את כל המשתנים של השפה העילית באותם שמות (כולל הפרמטרים).
3. תאתחלו את הפרמטרים ע"י העתקה מהמחסנית של הערכים.
4. תאתחלו את המשתנים המקומיים 
5. תממשו את הקוד הביצועי של התוכנית לפי המבנים שנלמדו בקורס, מומלץ לפי המימושים האינטואטיביים.
6. אחרי הפיתוח והדיבוג של התוכנית, אפשר להכניס יעול, כמו ביטול משתנים לטובת אוגרים או מימושים יעילים של לולאות.
7. הסבה למודל Large אם נדרש.
דוגמא,
נניח שברצונינו לממש bublesort באסמבלי.
ננמש את האלגוריתם בשפה עיךית (C במקרה הזה):
void bublesort(int vec[], int n)            /*****/
{
  int i, j, flag = 1;

  for (i = 0; (i < n) && flag; i++)
    {
      flag = 0;
      for( j = n-1; j > i; j--)
	if ( vec[j] < vec[j-1] )           /******/
	  {
	    flag = 1;
	    swap(&vec[j], &vec[j-1]);         /*****/
	  }
    }
}





עכשיו נממש אותו באסמבלי:

1. תבנו את המסגרת של התוכנית:
;
;  abubs.asm
;
     .MODEL SMALL
     .DATA 
     .CODE
PUBLIC _bublesort
     .386
;
;  extern void bublesort(int vec[], int n);
;                         [BP+4]    [BP+6]
_bublesort PROC NEAR
   PUSH BP
   MOV BP,SP
   PUSH SI
   PUSH DI
;

;

   POP DI
   POP SI
   POP BP   
   RET
_bublesort  ENDP

     END


2. מימוש ב-DATA של כל המשתנים:
     .DATA
vec DW ?
n   DW ?
i DW ?
j DW ?
flag DW ?

3. אתחול משתני הפרמטר:

   MOV AX,[BP+4]
   MOV vec,AX
   MOV AX,[BP+6]
   MOV n,AX
;
4. אתחול משתנים אוטומטיים:
MOV flag,1

5. תרגום ידני אל יתר הקוד:
   MOV i,0
For1:
     MOV AX,i
     CMP AX,n
     JNL Endfor1
     CMP flag,1
     JNE Endfor1
;
     MOV flag,0
     
     MOV AX,n
     DEC AX
     MOV j,AX
For2:
     MOV AX,j
     CMP AX,i
     JNG Endfor2
If1:
     MOV SI,j
     ADD SI,SI ; si =2*j
     ADD SI,vec ; SI = &vec[j]
     MOV AX,[SI]
     CMP AX,[SI-2]    
     JNL Skip1
       MOV flag,1
       XCHG AX,[SI-2]
       MOV [SI],AX
Skip1:
     
      DEC j
      JMP For2
Endfor2:

      INC i
      JMP For1
Endfor1:

התוכנית המלאה:
;
;  abubs.asm
;
     .MODEL SMALL
     .DATA
vec DW ?
n   DW ?
i DW ?
j DW ?
flag DW ?

     .CODE
PUBLIC _bublesort
     .386
;
;  extern void bublesort(int vec[], int n);
;                         [BP+4]    [BP+6]
_bublesort PROC NEAR
   PUSH BP
   MOV BP,SP
   PUSH SI
   PUSH DI
;
   MOV AX,[BP+4]
   MOV vec,AX
   MOV AX,[BP+6]
   MOV n,AX
;
   MOV flag,1
   MOV i,0
For1:
     MOV AX,i
     CMP AX,n
     JNL Endfor1
     CMP flag,1
     JNE Endfor1
;
     MOV flag,0
     
     MOV AX,n
     DEC AX
     MOV j,AX
For2:
     MOV AX,j
     CMP AX,i
     JNG Endfor2
If1:
     MOV SI,j
     ADD SI,SI ; si =2*j
     ADD SI,vec ; SI = &vec[j]
     MOV AX,[SI]
     CMP AX,[SI-2]    
     JNL Skip1
       MOV flag,1
       XCHG AX,[SI-2]
       MOV [SI],AX
Skip1:
 
      DEC j
      JMP For2
Endfor2:
      INC i
      JMP For1
Endfor1:
;
   POP DI
   POP SI
   POP BP   
   RET
_bublesort  ENDP

     END     
המימוש של הפונקציה bublesort במודל Large יהיה כדלקמן:
;
;  abubsl.asm
;
     .MODEL LARGE
     .DATA
vec DD ?
n   DW ?
i DW ?
j DW ?
flag DW ?

     .CODE
PUBLIC _bublesort
     .386
;
;  extern void bublesort(int vec[], int n);
;                         [BP+6]    [BP+10]
_bublesort PROC FAR
   PUSH BP
   MOV BP,SP
   PUSH SI
   PUSH DI
   PUSH ES
;
   MOV EAX,[BP+6]
   MOV vec,EAX
   MOV AX,[BP+10]
   MOV n,AX
;
   MOV flag,1
   MOV i,0
For1:
     MOV AX,i
     CMP AX,n
     JNL Endfor1
     CMP flag,1
     JNE Endfor1
;
     MOV flag,0
     
     MOV AX,n
     DEC AX
     MOV j,AX
For2:
     MOV AX,j
     CMP AX,i
     JNG Endfor2
If1:
     MOV SI,j
     ADD SI,SI ; si =2*j
     ADD SI,WORD PTR vec ; SI = &vec[j]
     MOV ES,WORD PTR vec+2
     MOV AX,ES:[SI]
     CMP AX,ES:[SI-2]    
     JNL Skip1
       MOV flag,1
       XCHG AX,ES:[SI-2]
       MOV ES:[SI],AX
Skip1:
     
      DEC j
      JMP For2
Endfor2:

      INC i
      JMP For1
Endfor1:
;
   POP ES
   POP DI
   POP SI
   POP BP   
   RET
_bublesort  ENDP

     END     

אם נצליב את שני המימושים ונראה את ההבדלים:

  Smallמודל                                                                                                                 Large  מודל  
;                                                                                                                ;
;  abubs.asm                                                                                           ;  abubsl.asm
;                                                                                                                ;
     .MODEL SMALL                                                                                             .MODEL LARGE
     .DATA                                                                                                                   .DATA
vec DW ?                                                                                                  vec DD ?
n   DW ?                                                                                                    n   DW ?
i DW ?                                                                                                        i DW ?
j DW ?                                                                                                        j DW ?
flag DW ?                                                                                                  flag DW ?
     .CODE                                                                                                                        .CODE
PUBLIC _bublesort                                                                                  PUBLIC _bublesort
     .386                                                                                                                               .386
;                                                                                                            ;
;  extern void bublesort(int vec[], int n);                                       ;  extern void bublesort(int vec[], int n);
;                                         [BP+4]    [BP+6]                                       ;                                        [BP+6]    [BP+10]
_bublesort PROC NEAR                                                                            _bublesort PROC FAR
   PUSH BP                                                                                                            PUSH BP
   MOV BP,SP                                                                                                       MOV BP,SP
   PUSH SI                                                                                                             PUSH SI
   PUSH DI                                                                                                            PUSH DI
                                                                                                                              PUSH ES
;                                                                                                            ;
   MOV AX,[BP+4]                                                                                               MOV EAX,[BP+6]
   MOV vec,AX                                                                                                     MOV vec,EAX
   MOV AX,[BP+6]                                                                                               MOV AX,[BP+10]
   MOV n,AX                                                                                                        MOV n,AX
;                                                                                                            ;
   MOV flag,1                                                                                                       MOV flag,1
   MOV i,0                                                                                                             MOV i,0
For1:                                                                                                For1:
     MOV AX,I                                                                                                        MOV AX,i
     CMP AX,n                                                                                                        CMP AX,n
     JNL Endfor1                                                                                                    JNL Endfor1
     CMP flag,1                                                                                                      CMP flag,1
     JNE Endfor1                                                                                                    JNE Endfor1
;                                                                                                         ;
     MOV flag,0                                                                                                      MOV flag,0 
     MOV AX,n                                                                                                       MOV AX,n
     DEC AX                                                                                                            DEC AX 
     MOV j,AX                                                                                                        MOV j,AX
For2:                                                                                                  For2:
     MOV AX,j                                                                                                        MOV AX,j
     CMP AX,I                                                                                                         CMP AX,i
     JNG Endfor2                                                                                                   JNG Endfor2
If1:                                                                                                    If1:
     MOV SI,j                                                                                                          MOV SI,j
     ADD SI,SI ; si =2*j                                                                                           ADD SI,SI ; si =2*j
     ADD SI,vec ; SI = &vec[j]                                                                               ADD SI,WORD PTR vec
                                                                                                                                                ;SI=&vec[j]
                                                                                                                               MOV ES, WORD PTR vec+2 
[bookmark: _GoBack]    MOV AX,[SI]                                                                                                     MOV AX,ES:[SI]
     CMP AX,[SI-2]                                                                                                 CMP AX,ES:[SI-2]    
     JNL Skip1                                                                                                         JNL Skip1
       MOV flag,1                                                                                                        MOV flag,1
       XCHG AX,[SI-2]                                                                                                  XCHG AX,ES:[SI-2]
       MOV [SI],AX                                                                                                       MOV ES:[SI],AX
Skip1:                                                                                                Skip1:
      DEC j                                                                                                               DEC j
      JMP For2                                                                                                        JMP For2
Endfor2:                                                                                           Endfor2
      INC I                                                                                                                 INC i
      JMP For1                                                                                                         JMP For1
Endfor1:                                                                                                          Endfor1:
;
                                                                                                        ;
                                                                                                                        POP ES
   POP DI                                                                                                         POP DI
   POP SI                                                                                                          POP SI
   POP BP                                                                                                         POP BP   
   RET                                                                                                               RET
_bublesort  ENDP                                                                          bublesort  ENDP
                                        
     END                                                                                                          END     
